Artificial intelligence and the skill premium

Jamel Saadaoui, University of Strasbourg, BETA, CNRS Posted 1 year ago

ORIGINAL ARTICLE: David E. Bloom, Klaus Prettner, Jamel Saadaoui, Mario Veruete (2023), Artificial intelligence and the skill premium, arXiv:2311.09255. DOI: https://doi.org/10.48550/arXiv.2311.09255

What will likely be the effect of the emergence of ChatGPT and other forms of artificial intelligence (AI) on the skill premium?

To understand the consequences of ChatGPT and other forms of artificial intelligence, we developed a nested CES production function that distinguishes between industrial robots and AI. Industrial robots predominantly substitute for low-skill workers, whereas AI mainly helps to perform the tasks of high-skill workers. We show that AI reduces the skill premium as long as it is more substitutable for high-skill workers than low-skill workers are for high-skill workers.

Section 1 provides the notations and the assumptions that will be required for the calculations. Section 2 introduces the production function and a more readable notation. Then, Section 3 computes the wage rates for low- and high-skilled workers thanks to partial derivatives. Section 4 provides the wage-skill premium. Sections 5 and 6 calculate the partial derivatives of the different wage rates relative to industrial robots and AI. Section 7 provides interactive 2-D and 3-D plots to visualize the evolutions of the wage skill premium for different values of the parameters and the variables.

This notebook aims to replicate all the computations in the paper “Artificial intelligence and the skill premium” (https://arxiv.org/abs/2311.09255)

You can freely download a copy of this notebook to your own computer or make a copy of it to your Wolfram Cloud account (it is possible to create a free account).

This notebook contains "Initialization Cells" we recommend to say "yes" when asked to evaluate them.

You can also evaluate this notebook step by step or simply by clicking on the menu **Evaluation** then in **Evaluate Notebook**.

1. Production function Y_t.
2. Computation of the wage rates $w_{u,t} = \frac{\partial Y}{\partial L_u}$ and $w_{s,t} = \frac{\partial Y}{\partial L_s}$.
3. Computation of the skill premium $= \frac{W_s}{w_u}$.
4. Computation of $\frac{\partial w_s}{\partial \mathcal{G}}, \frac{\partial w_s}{\partial P}$.
5. Computation of $\frac{\partial w_u}{\partial \mathcal{G}}, \frac{\partial w_u}{\partial P}$.
6. Plots

---

1 Notation and Assumptions »

---

2 Production function Y_t

- P[t] = industrial robots
- $L_s[t]$ = skilled workers
- $L_u[t]$ = unskilled workers

- $\mathcal{G}$[t] = AI
- $\mathcal{K}$[t] = capital stocks, machines, etc.

```
In[∘]:= Clear[productionFunctionYt, productionFunctionYtM];
```

$$\text{productionFunctionYtM} = (\text{M}[\text{L}_u[t], \text{L}_s[t], \text{P}[t], \mathcal{G}[t]])^{\frac{1-\alpha}{\gamma}}\, \mathcal{K}[t]^\alpha;$$

$$\text{productionFunctionYt} = \left(\beta[3]\left(\beta[1]\,(\text{L}_u[t])^\theta + (1-\beta[1])\,(\text{P}[t])^\theta\right)^{\gamma/\theta} + (1-\beta[3])\left(\beta[2]\,(\text{L}_s[t])^\varphi + (1-\beta[2])\,(\mathcal{G}[t])^\varphi\right)^{\gamma/\varphi}\right)^{\frac{1-\alpha}{\gamma}} \mathcal{K}[t]^\alpha$$

In the following, for readability purposes, we introduce the notation:

$$M = \beta_3\left(\beta_1\,(L_u[t])^\theta + (1-\beta_1)\,(P[t])^\theta\right)^{\gamma/\theta} + (1-\beta_3)\left(\beta_2\,(L_s[t])^\varphi + (1-\beta_2)\,(\mathcal{G}[t])^\varphi\right)^{\gamma/\varphi}.$$

```
In[∘]:= TraditionalForm[productionFunctionYtM /. MTrule /. Trule]
```

Out[∘]//TraditionalForm=

$$\mathcal{K}^\alpha M^{\frac{1-\alpha}{\gamma}}$$

```
In[∘]:= TraditionalForm[productionFunctionYt /. Trule]
```

Out[∘]//TraditionalForm=

$$\mathcal{K}^\alpha\left(\beta_3\left(\beta_1 L_u^\theta + (1-\beta_1) P^\theta\right)^{\gamma/\theta} + (1-\beta_3)\left((1-\beta_2)\mathcal{G}^\varphi + \beta_2 L_s^\varphi\right)^{\gamma/\varphi}\right)^{\frac{1-\alpha}{\gamma}}$$

---

3 Computation of the wage rates $w_{u,t} = \dfrac{\partial Y}{\partial L_u}$ and $w_{s,t} = \dfrac{\partial Y}{\partial L_s}$

Wage unskilled w_u

```
In[∘]:= wageRateUShort = D[productionFunctionYtM, Lu[t]];
        TraditionalForm[
          wageRateUShort /. MTrule /. Trule
          ]
```

Out[∘]//TraditionalForm=

$$\frac{(1-\alpha)\mathcal{K}^\alpha M^{\frac{1-\alpha}{\gamma}-1} M^{(1,0,0,0)}(L_u, L_s, P, \mathcal{G})}{\gamma}$$

NB. In Wolfram Language, Derivative[{n_1, n_2, …}][f] represents the derivative of f[{x_1, x_2, …}] taken n_i times with respect t
given in lists in f can be handled by using a corresponding list structure in Derivative.

If $f = f(x_1, x_2, \ldots, x_i, \ldots, x_n)$, then, the first order partial derivative $\frac{\partial f}{x_i}$ is given by:

$$\frac{\partial f}{x_i}(x_1, x_2, \ldots, x_i, \ldots, x_n) = \text{Derivative}[0, 0, \ldots, 1, 0, \ldots, 0][f][x_1, x_2, \ldots, x_i, \ldots, x_n] = f^{(0,0,\ldots,1,0\ldots,0)}(x_1, x_2, \ldots, x_i, \ldots, x_n).$$

In particular, $M^{(0,1,0,0)}(L_u, L_s, P, \mathcal{G})$ represents $\frac{\partial M}{\partial L_s}$. Similarly, $M^{(0,1,0,1)}(L_u, L_s, P, \mathcal{G})$ represents $\frac{\partial^2 M}{\partial L_s \partial \mathcal{G}}$.

For more details, the documentation page of the function Derivative.

```
In[∘]:= wageRateU = D[productionFunctionYt, Lu[t]];
        TraditionalForm[wageRateU /. Trule]
```

Out[∘]//TraditionalForm=

$$(1-\alpha)\beta_1\beta_3\mathcal{K}^\alpha L_u^{\theta-1}\left(\beta_1 L_u^\theta + (1-\beta_1)P^\theta\right)^{\frac{\gamma}{\theta}-1}\left(\beta_3\left(\beta_1 L_u^\theta + (1-\beta_1)P^\theta\right)^{\gamma/\theta} + (1-\beta_3)\left((1-\beta_2)\mathcal{G}^\varphi + \beta_2 L_s^\varphi\right)^{\gamma/\varphi}\right)^{\frac{1-\alpha}{\gamma}-1}$$

Wage skilled w_s

```
In[∘]:= wageRateSShort = D[productionFunctionYtM, Ls[t]];
        TraditionalForm[
          wageRateSShort /. MTrule /. Trule
          ]
```

Out[∘]//TraditionalForm=

$$\frac{(1-\alpha)\mathcal{K}^\alpha M^{\frac{1-\alpha}{\gamma}-1} M^{(0,1,0,0)}(L_u, L_s, P, \mathcal{G})}{\gamma}$$

In[∘]:= wageRateS = D[productionFunctionYt, L$_s$[t]];
TraditionalForm[wageRateS /. Trule]

Out[∘]//TraditionalForm=

$$(1-\alpha)\beta_2(1-\beta_3)\mathcal{K}^\alpha L_s^{\varphi-1}\left((1-\beta_2)\mathcal{G}^\varphi+\beta_2 L_s^\varphi\right)^{\frac{\gamma}{\varphi}-1}\left(\beta_3\left(\beta_1 L_u^\theta+(1-\beta_1)P^\theta\right)^{\gamma/\theta}+(1-\beta_3)\left((1-\beta_2)\mathcal{G}^\varphi+\beta_2 L_s^\varphi\right)^{\gamma/\varphi}\right)^{\frac{1-\alpha}{\gamma}-1}$$

4 Computation of the skill premium $\frac{w_s}{w_u}$.

In[∘]:= skillPremiumShort = $\frac{\text{wageRateSShort}}{\text{wageRateUShort}}$;

TraditionalForm[
skillPremiumShort /. Trule
]

Out[∘]//TraditionalForm=

$$\frac{M^{(0,1,0,0)}(L_u, L_s, P, \mathcal{G})}{M^{(1,0,0,0)}(L_u, L_s, P, \mathcal{G})}$$

In[∘]:= skillPremium = $\frac{\text{wageRateS}}{\text{wageRateU}}$;

TraditionalForm[skillPremium /. Trule]

Out[∘]//TraditionalForm=

$$\frac{\beta_2(1-\beta_3)L_s^{\varphi-1}L_u^{1-\theta}\left(\beta_1 L_u^\theta+(1-\beta_1)P^\theta\right)^{1-\frac{\gamma}{\theta}}\left((1-\beta_2)\mathcal{G}^\varphi+\beta_2 L_s^\varphi\right)^{\frac{\gamma}{\varphi}-1}}{\beta_1\beta_3}$$

5 Computation of $\frac{\partial w_s}{\partial G_t}, \frac{\partial w_s}{\partial P_t}$.

$\frac{\partial w_s}{\partial G_t}$

In[∘]:= TraditionalForm[D[wageRateSShort, $\mathcal{G}$[t]] /. Trule]

Out[∘]//TraditionalForm=

$$\frac{(1-\alpha)\left(\frac{1-\alpha}{\gamma}-1\right)\mathcal{K}^\alpha M^{(0,0,0,1)}(L_u, L_s, P, \mathcal{G})M^{(0,1,0,0)}(L_u, L_s, P, \mathcal{G})M(L_u, L_s, P, \mathcal{G})^{\frac{1-\alpha}{\gamma}-2}}{\gamma}+\frac{(1-\alpha)\mathcal{K}^\alpha M^{(0,1,0,1)}(L_u, L_s, P, \mathcal{G})M\ldots}{\gamma}$$

In[∘]:= TraditionalForm[D[wageRateS, $\mathcal{G}$[t]] /. Trule]

Out[∘]//TraditionalForm=

$$(1-\alpha)(1-\beta_2)\beta_2(1-\beta_3)^2\gamma\left(\frac{1-\alpha}{\gamma}-1\right)\mathcal{G}^{\varphi-1}\mathcal{K}^\alpha L_s^{\varphi-1}\left((1-\beta_2)\mathcal{G}^\varphi+\beta_2 L_s^\varphi\right)^{\frac{2\gamma}{\varphi}-2}\left(\beta_3\left(\beta_1 L_u^\theta+(1-\beta_1)P^\theta\right)^{\gamma/\theta}+(1-\beta_3)\left((1-\beta_2)\ldots\right.\right.$$

$$\beta_2)\beta_2(1-\beta_3)\varphi\left(\frac{\gamma}{\varphi}-1\right)\mathcal{G}^{\varphi-1}\mathcal{K}^\alpha L_s^{\varphi-1}\left((1-\beta_2)\mathcal{G}^\varphi+\beta_2 L_s^\varphi\right)^{\frac{\gamma}{\varphi}-2}\left(\beta_3\left(\beta_1 L_u^\theta+(1-\beta_1)P^\theta\right)^{\gamma/\theta}+(1-\beta_3)\left((1-\beta_2)\mathcal{G}^\varphi+\beta_2 L_s^\varphi\right)^{\gamma/\ldots}\right.$$

$\frac{\partial w_s}{\partial P_t}$

In[∘]:= TraditionalForm[D[wageRateSShort, P[t]] /. Trule]

Out[∘]//TraditionalForm=

$$\frac{(1-\alpha)\left(\frac{1-\alpha}{\gamma}-1\right)\mathcal{K}^\alpha M^{(0,0,1,0)}(L_u, L_s, P, \mathcal{G})M^{(0,1,0,0)}(L_u, L_s, P, \mathcal{G})M(L_u, L_s, P, \mathcal{G})^{\frac{1-\alpha}{\gamma}-2}}{\gamma}+\frac{(1-\alpha)\mathcal{K}^\alpha M^{(0,1,1,0)}(L_u, L_s, P, \mathcal{G})M\ldots}{\gamma}$$

In[∘]:= TraditionalForm[D[wageRateS, P[t]]]

Out[∘]//TraditionalForm=

$$(1-\alpha)(1-\beta_1)\beta_2(1-\beta_3)\beta_3\gamma\left(\frac{1-\alpha}{\gamma}-1\right)P(t)^{\theta-1}\mathcal{K}(t)^\alpha L_s(t)^{\varphi-1}\left(\beta_1 L_u(t)^\theta+(1-\beta_1)P(t)^\theta\right)^{\frac{\gamma}{\theta}-1}\left(\beta_2 L_s(t)^\varphi+(1-\beta_2)\mathcal{G}(t)^\varphi\right)^{\frac{\gamma}{\varphi}-\ldots}$$

$$\left(\beta_3\left(\beta_1 L_u(t)^\theta+(1-\beta_1)P(t)^\theta\right)^{\gamma/\theta}+(1-\beta_3)\left(\beta_2 L_s(t)^\varphi+(1-\beta_2)\mathcal{G}(t)^\varphi\right)^{\gamma/\varphi}\right)^{\frac{-}{\gamma}-2}$$

6 Computation of $\frac{\partial w_u}{\partial G_t}, \frac{\partial w_u}{\partial P_t}$

$\frac{\partial w_u}{\partial G_t}$

In[∘]:= **TraditionalForm[D[wageRateUShort, $\mathcal{G}$[t]] /. Trule]**

Out[∘]//TraditionalForm=

$$\frac{(1-\alpha)\left(\frac{1-\alpha}{\gamma}-1\right)\mathcal{K}^\alpha M^{(0,0,0,1)}(L_u,L_s,P,\mathcal{G})\,M^{(1,0,0,0)}(L_u,L_s,P,\mathcal{G})\,M(L_u,L_s,P,\mathcal{G})^{\frac{1-\alpha}{\gamma}-2}}{\gamma}+\frac{(1-\alpha)\mathcal{K}^\alpha M^{(1,0,0,1)}(L_u,L_s,P,\mathcal{G})\,M}{\gamma}$$

In[∘]:= **TraditionalForm[D[wageRateU, $\mathcal{G}$[t]]]**

Out[∘]//TraditionalForm=

$$(1-\alpha)\beta_1(1-\beta_2)(1-\beta_3)\beta_3\gamma\left(\frac{1-\alpha}{\gamma}-1\right)\mathcal{G}(t)^{\varphi-1}\mathcal{K}(t)^\alpha L_u(t)^{\theta-1}\left(\beta_1 L_u(t)^\theta+(1-\beta_1)P(t)^\theta\right)^{\frac{\gamma}{\theta}-1}\left(\beta_2 L_s(t)^\varphi+(1-\beta_2)\mathcal{G}(t)^\varphi\right)^{\frac{\gamma}{\varphi}-}$$

$$\left(\beta_3\left(\beta_1 L_u(t)^\theta+(1-\beta_1)P(t)^\theta\right)^{\gamma/\theta}+(1-\beta_3)\left(\beta_2 L_s(t)^\varphi+(1-\beta_2)\mathcal{G}(t)^\varphi\right)^{\gamma/\varphi}\right)^{\frac{1-\alpha}{\gamma}-2}$$

$\frac{\partial w_u}{\partial P_t}$

In[∘]:= **TraditionalForm[D[wageRateUShort, P[t]] /. Trule]**

Out[∘]//TraditionalForm=

$$\frac{(1-\alpha)\left(\frac{1-\alpha}{\gamma}-1\right)\mathcal{K}^\alpha M^{(0,0,1,0)}(L_u,L_s,P,\mathcal{G})\,M^{(1,0,0,0)}(L_u,L_s,P,\mathcal{G})\,M(L_u,L_s,P,\mathcal{G})^{\frac{1-\alpha}{\gamma}-2}}{\gamma}+\frac{(1-\alpha)\mathcal{K}^\alpha M^{(1,0,1,0)}(L_u,L_s,P,\mathcal{G})\,M}{\gamma}$$

In[∘]:= **TraditionalForm[D[wageRateU, P[t]]]**

Out[∘]//TraditionalForm=

$$(1-\alpha)(1-\beta_1)\beta_1\beta_3^2\gamma\left(\frac{1-\alpha}{\gamma}-1\right)P(t)^{\theta-1}\mathcal{K}(t)^\alpha L_u(t)^{\theta-1}\left(\beta_1 L_u(t)^\theta+(1-\beta_1)P(t)^\theta\right)^{\frac{2\gamma}{\theta}-2}$$

$$\left(\beta_3\left(\beta_1 L_u(t)^\theta+(1-\beta_1)P(t)^\theta\right)^{\gamma/\theta}+(1-\beta_3)\left(\beta_2 L_s(t)^\varphi+(1-\beta_2)\mathcal{G}(t)^\varphi\right)^{\gamma/\varphi}\right)^{\frac{1-\alpha}{\gamma}-2}+(1-\alpha)(1-\beta_1)\beta_1\beta_3\theta\left(\frac{\gamma}{\theta}-1\right)P(t)^{\theta-1}\mathcal{K}(t$$

$$\left(\beta_1 L_u(t)^\theta+(1-\beta_1)P(t)^\theta\right)^{\frac{\gamma}{\theta}-2}\left(\beta_3\left(\beta_1 L_u(t)^\theta+(1-\beta_1)P(t)^\theta\right)^{\gamma/\theta}+(1-\beta_3)\left(\beta_2 L_s(t)^\varphi+(1-\beta_2)\mathcal{G}(t)^\varphi\right)^{\gamma/\varphi}\right)^{\frac{1-\alpha}{\gamma}-1}$$

7 Plots

In[∘]:=
```
parametersValues = {α → 1/3, γ → 1/3, θ → 3/4, φ → 1/2, β1 → 0.9, β2 → 0.95, β3 → 2/3};
stateVariablesValues = {
  𝒦[t] -> 69042181800420,
  Lu[t] -> 98299000,
  Ls[t] -> 58446000,
  P[t] -> 17282199580
  };
```

If you are in the cloud, in order to see the following plots you should open the file as a copy on your cloud account OR download a copy of the file and open it with Mathematica or with WolframPlayer (for those who does not have access to Mathematica).

In[∘]:=
```
DynamicModule[
  {expressionToPlot, select},
  Manipulate[
   expressionToPlot = ReplaceAll[
     select[functionToPlot] /. Join[parametersValues, stateVariablesValues],
     f_[t_] :> f
```

```
      f_[t_] :> f
      ];
    Plot[
      Evaluate[expressionToPlot], {𝒢, 0, 2 (17282199580) (* 2 𝒢t, cf. Table 2 of the paper *) },
      PlotLabel -> functionToPlot <> " as a function of 𝒢",
      PlotLegends -> Placed[TraditionalForm[select[functionToPlot] /. Trule], Below],
      PlotTheme -> "Detailed",
      PlotStyle -> {Thick, color[Position[{"ws", "wu", "ws / wu"}, functionToPlot][[1, 1]]]},
      FrameLabel -> {"𝒢", functionToPlot},
      ImageSize -> Large
      ],
    {{functionToPlot, "ws", "Select a function"}, {"ws", "wu", "ws / wu"}},
    Initialization :> {
      select[s_String] := Which[s == "ws", wageRateS, s == "wu", wageRateU, s == "ws/wu", skillPremium],
      color[i_] := Part[{RGBColor[0.17, 0.19, 0.97], RGBColor[0.1, 0.67, 0.1], RGBColor[0.98, 0.75, 0.14], RGBColor[0.98, 
              0.41, 0.93]}, i]
      },
    SaveDefinitions -> True
    ]
  ]
```




```
In[∘]:= stateVariablesValues2 = {
    𝒦[t] -> 69042181800420,
    Lu[t] -> 98299000,
    Ls[t] -> 58446000
    };
```

```
In[∘]:= DynamicModule[
    {expressionToPlot, select},
    Manipulate[
      expressionToPlot = ReplaceAll[
        select[functionToPlot] /. Join[parametersValues, stateVariablesValues2],
        f_[t_] :> f
        ];
      Show[
        Plot3D[
          Evaluate[expressionToPlot],
          {𝒢, 0, 2 (17282199580) (* 2 𝒢t, cf. Table 2 of the paper *) },
          {P, 0, 2 (17282199580) (* 2 𝒢t, cf. Table 2 of the paper *) },
          BoxRatios -> {1, 1, 2 / 3},
```

```
        PlotPoints -> 50,
        MaxRecursion -> 4,
        ColorFunction -> "TemperatureMap",
        Mesh -> None,
        (*Mesh -> {5, 5, 15},
          MeshFunctions -> {#1 &, #2 &, #3 &},
          MeshStyle -> {{Gray}, {Gray}, {Purple}}, *)
        PlotLabel -> functionToPlot <> "(𝒢,P)",
        (*PlotLegends -> Placed[TraditionalForm[select[functionToPlot] /. Trule], Below], *)
        PlotTheme -> "Detailed",
        AxesLabel -> {"𝒢", "P", functionToPlot},
        ImageSize -> Medium
        ],
      Graphics3D[
        {
          {Red, PointSize[0.03], Point[{𝒢, P, expressionToPlot} /. {𝒢 -> 0, P -> 2 (17282199580)}]},
          {Blue, PointSize[0.03], Point[{𝒢, P, expressionToPlot} /. {𝒢 -> 2 (17282199580), P -> 2 (17282199580)}]}
          }
        ],
      ParametricPlot3D[{gslice, P, expressionToPlot} /. {𝒢 -> gslice}, {P, 0, 2 (17282199580)}, PlotStyle -> {Thick, Blac
      ParametricPlot3D[{𝒢, pslice, expressionToPlot} /. {P -> pslice}, {𝒢, 0, 2 (17282199580)}, PlotStyle -> {Thick, Gra
      ],
    {{functionToPlot, "w_s", "Select a function"}, {"w_s", "w_u", "w_s/w_u"}},
    {{gslice, (17282199580), "𝒢 slice"}, 0, 2 (17282199580)},
    {{pslice, (17282199580), "P slice"}, 0, 2 (17282199580)},
    Initialization :> {
      select[s_String] := Which[s == "w_s", wageRateS, s == "w_u", wageRateU, s == "w_s/w_u", skillPremium]
      },
    SaveDefinitions -> True
    ]
  ]
```




```
In[•]:= DynamicModule[
  {expressionToPlot, select},
  Manipulate[
    expressionToPlot = ReplaceAll[
      select[functionToPlot] /. Join[parametersValues, stateVariablesValues2],
      f_[t_] :> f
      ];
    Show[
```

```
Show[
  ContourPlot[
    Evaluate[expressionToPlot],
    {𝒢, 0, 2 (17282199580) (* 2 𝒢_t, cf. Table 2 of the paper *) },
    {P, 0, 2 (17282199580) (* 2 𝒢_t, cf. Table 2 of the paper *) },
    BoxRatios -> {1, 1, 2 / 3},
    PlotPoints -> 50,
    MaxRecursion -> 4,
    ColorFunction -> "TemperatureMap",
    Mesh -> {0, 0, 0},
    MeshFunctions -> {#1 &, #2 &, #3 &},
    MeshStyle -> {{LightGray}, {LightGray}, {Purple}},
    PlotLabel -> functionToPlot <> "(𝒢,P)",
    (* PlotLegends -> Placed[TraditionalForm[select[functionToPlot] /. Trule], Below], *)
    PlotTheme -> "Detailed",
    FrameLabel -> {"𝒢", "P"},
    ImageSize -> Medium,
    FrameTicks -> {{0.5 * 10^10, 1.5 * 10^10, 2.5 * 10^10, 3.5 * 10^10}, Automatic}
    ],
  VectorPlot[
    Evaluate[Grad[expressionToPlot, {𝒢, P}]],
    {𝒢, 0, 2 (17282199580) (* 2 𝒢_t, cf. Table 2 of the paper *) },
    {P, 0, 2 (17282199580) (* 2 𝒢_t, cf. Table 2 of the paper *) },
    VectorPoints -> 10,
    VectorColorFunction -> "TemperatureMap",
    VectorScaling -> "Linear"
    ],
  Graphics[
    {
      {Red, PointSize[0.03], Point[{𝒢, P} /. {𝒢 -> 0, P -> 2 (17282199580)}]},
      {Blue, PointSize[0.03], Point[{𝒢, P} /. {𝒢 -> 2 (17282199580), P -> 2 (17282199580)}]}
      }
    ],
  ParametricPlot[{gslice, P} /. {𝒢 -> gslice}, {P, 0, 2 (17282199580)}, PlotStyle -> {Thick, Black}],
  ParametricPlot[{𝒢, pslice} /. {P -> pslice}, {𝒢, 0, 2 (17282199580)}, PlotStyle -> {Thick, Gray}]
  ],
 {{functionToPlot, "w_s", "Select a function"}, {"w_s", "w_u", "w_s / w_u"}},
 {{gslice, (17282199580), "𝒢 slice"}, 0, 2 (17282199580)},
 {{pslice, (17282199580), "P slice"}, 0, 2 (17282199580)},
 Initialization :> {
   select[s_String] := Which[s == "w_s", wageRateS, s == "w_u", wageRateU, s == "w_s / w_u", skillPremium]
   },
 SaveDefinitions -> True
 ]
]
```





CITE THIS NOTEBOOK

Artificial intelligence and the skill premium
by David Bloom, Klaus Prettner, Jamel Saadaoui, Mario Veruete
Wolfram Community, STAFF PICKS, November 21, 2023
https://community.wolfram.com/groups/-/m/t/3068735

Get this Notebook Powered by Wolfram

Attachments: DOWLOAD-DESKTOP-...nb

Reply | Flag

1 Reply

Sort By: Replies Likes Recent

0



EDITORIAL BOARD, WOLFRAM
Posted 1 year ago

-- you have earned ***Featured Contributor Badge*** Your exceptional post has been selected for our editorial columns ***Staff Picks*** http://wolfr.am/StaffPicks and ***Publication Materials*** https://wolfr.am/PubMat and Your Profile is now distinguished by a ***Featured Contributor Badge*** and is displayed on the Featured Contributor Board. Thank you!

Reply | Flag

Feedback

Reply to this discussion

Add Notebook

Reply Preview

Attachments

Add a file to this post

Follow this discussion

Publish or Discard